

Numerische Lösungen mathematischer Probleme

Wir stellen Ihnen eine kleine Sammlung von mathematischen Unterprogrammen vor, die in eigenen Programmen benötigt werden, aber schwer zu programmieren sind.

In dem ersten Programm (Listing 1) wird näherungsweise die erste Ableitung einer Funktion an einer bestimmten Stelle mit Hilfe des Differenzenquotienten bestimmt. Der Differenzenquotient ist wie folgt definiert:

$$f'(z) = \frac{f(x) - f(x_0)}{x - x_0} \quad \text{mit } x_0 \leq z \leq x$$

Dieser Differenzenquotient (im Programm Zeile 200) wird etwas abgewandelt, so daß man den Abstand zwischen x und x_0 immer kleiner werden läßt. Dies geschieht, indem man statt x den Wert $x_0 + h$ nimmt. Der Wert h wird dann immer weiter halbiert (Zeile 230), bis eine vorgegebene Genauigkeit erreicht ist (Zeile 220). Als erste Ableitung der bestimmten Stelle ist dann der letzte ausgedruckte Wert zu nehmen. Die erste Ableitung gibt übrigens die Steigung der Tangenten an, die an der bestimmten Stelle an den Graphen der Funktion angelegt wird.

Diese Tangente, beziehungsweise auch weitere Tangenten werden bei dem zweiten und dritten Programm (Listing 2 und 3) benötigt. Bei dem zweiten Programm wird mit Hilfe von Tangenten die Nullstelle(n) (die Stelle, wo der Graph der Funktion die x -Achse schneidet) einer Funktion bestimmt.

Die Iterationsvorschrift (Zeile 160) lautet:

$$x_{i+1} = x_i - y_i / y'_i = x_i - f(x_i) / f'(x_i)$$

Dieses Verfahren stammt von Newton. Im dritten Programm (Listing 3) ist die Vorschrift ähnlich, nur braucht man die erste Ableitung nicht wissen, was oft von Vorteil ist. Die erste Ableitung wird hier in etwa so bestimmt, wie es in Programm 1 beschrieben ist. Da nicht nur Ableitungen und Nullstellen von Interesse sind, sondern auch die oben erwähnten Integrale, behandeln Programm 4 und 5 (Listing 4 und 5) diese Probleme.

Im vierten Programm (Listing 4) ist die Kepler'sche Faßregel umgesetzt. Möchte man das Integral einer Funktion (Zeile 240) wissen, so kann man es mit der Kepler'schen Faßregel annähern. Die Faßregel lautet:

$$\int_a^b f(x) dx \approx (b-a) \cdot (f(a)/6 + f((a+b)/2) \cdot 2/3 + f(b)/6)$$

Mit dieser noch recht einfachen Formel läßt sich das Integral schnell näherungsweise bestimmen.

Die Formel im Programm 5 (Listing 5) stammt von Simpson, sie ist ein wenig komplizierter:

$$\int_a^b f(x) dx \approx h/3 \cdot (f(a) + f(b) + 4 \cdot \sum_{k=0}^{n-1} f(a + (2k+1)h) + 2 \cdot \sum_{k=1}^{n-1} f(a + 2kh))$$

mit $h = (b-a)/(2n)$

Die beiden Summen sind in den Zeilen 210 bis 240 und 250 bis 280 umgesetzt.

Ein Problem aus einem etwas anderen Bereich bezieht sich auf gewöhnliche Differentialgleichungen. Bei dem Eulerschen Polygonzugverfahren in Programm 6 (Listing 6) wird der Graph, der die Lösung der gewöhnlichen Differentialgleichung beschreibt, in Form von kleinen Geradenstückchen genähert. Diese kleinen Geradenstückchen bilden dann den Polygonzug. Die einzelnen Punkte zu dieser Näherung wer-

den auf einem vorgegebenen Bereich bezüglich eines bestimmten Anfangswertes vorgegeben.

Das Verfahren von Runge-Kutta in Programm 7 (Listing 7) ist weitaus genauer, aber dafür ist die Formel auch komplizierter (Zeile 200, Zeilen 250 bis 280):

Anfangswert $y_0 = y(x_0)$

Schrittfunktion (Schrittweite h):

$$y_{k+1} = y_k + h/6 \cdot (L_{k,1} + 2L_{k,2} + 2L_{k,3} + L_{k,4})$$

mit $L_{k,1} = f(x_k, y_k)$ {mit $y' = f(x, y)$ }

$$L_{k,2} = f(x_k + h/2, y_k + h/2 \cdot L_{k,1})$$

$$L_{k,3} = f(x_k + h/2, y_k + h/2 \cdot L_{k,2})$$

$$L_{k,4} = f(x_k + h, y_k + h \cdot L_{k,3})$$

(Dietmar Rabich/ah)

```

100 REM DIFFERENZENQUOTIENT          <009>
110 PRINT "DIFFERENZENQUOTIENT"      <207>
120 INPUT "DIFFERENTIATIONSSTELLE"; X0 <202>
130 INPUT "STARTSTELLE {11SPACE}"; X1 <180>
140 IF X0=X1 THEN 120                 <213>
150 REM BERECHNUNG UND AUSGABE       <019>
160 PRINT "X", "F'"                 <024>
170 H=X1-X0                           <081>
180 X=X0+H:GOSUB 270:F0=F             <010>
190 X=X0:GOSUB 270:F1=F               <029>
200 DF=(F0-F1)/H                       <011>
210 PRINT INT((X0+H)*1E5+.5)/1E5,INT(DF*1E5+.5)/1E5 <050>
220 IF HK>(X1-X0)AND ABS(DF-DG)<1E-6 THEN <232>
    250                                     <240>
230 DG=DF:H=H/2                       <040>
240 GOTO 180                           <219>
250 PRINT "F' (" ; X0 ; ")=" ; DF      <008>
260 END                               <186>
270 REM FUNKTION                     <097>
280 F=X*X*X-2*X*X-2                  <094>
290 RETURN

```

0 64'er

```

DIFFERENZENQUOTIENT
DIFFERENTIATIONSSTELLE 1
STARTSTELLE 2
X      F'
2      1
1.5    -.25
1.25    -.6875
1.125    -.85937
1.0625    -.93359
1.03125    -.96777
1.01563    -.98413
1.00781    -.99213
1.00391    -.99608
1.00195    -.99804
1.00098    -.99902
1.00049    -.99951
1.00024    -.99976
1.00012    -.99988
1.00006    -.99994
1.00003    -.99997
1.00002    -1
1.00001    -1
F' ( 1 ) = -1

```

Listing 1. Differentialquotient mit einem Beispiel. Bitte verwenden Sie zum Eingeben den Checksummer auf Seite 6.

```

100 REM NULLSTELLENSUCHE NACH NEWTON <107>
110 PRINT"ITERATION NACH NEWTON" <040>
120 PRINT" X", " F" <094>
130 READ X0:REM STARTWERT <212>
140 X=X0:GOSUB 220:F0=F:REM FUNKTIONSWERT <136>
150 X=X0:GOSUB 250:F1=FS:REM 1. ABLEITUNG <221>
160 X1=X0-F0/F1 <066>
170 PRINT X1,F0 <019>
180 IF ABS(X1-X0)<1E-6 THEN END <166>
190 X0=X1 <034>
200 GOTO 140 <192>
210 REM FUNKTION <124>
220 F=X^2-4 <121>
230 RETURN <032>
240 REM 1. ABLEITUNG <021>
250 FS=2*X <117>
260 RETURN <064>
270 REM STARTWERT <223>
280 DATA 3 <193>

```

64'er

```

ITERATION NACH NEWTON
X      F
2.16666667      5.00000001
2.00641026      .694444446
2.00001024      .0256821178
2              4.09577042E-05
2              0

```

Listing 2. Nullstellen mit dem Newtonschen Iterationsverfahren.

```

100 REM NULLSTELLENSUCHE - REGULA-FALSI <028>
110 PRINT"REGULA FALSI" <175>
120 READ X0,X1:REM STARTWERTE <088>
130 PRINT" X", " F" <104>
140 X=X1:GOSUB 250:F1=F <007>
150 X=X0:GOSUB 250:F0=F <209>
155 PRINT X,F0 <089>
160 Z=X1-(X1-X0)/(F1-F0)*F1 <071>
180 X0=X1:X1=Z <139>
190 IF ABS(X1-X0)<1E-8 THEN 210 <089>
200 GOTO 140 <192>
210 REM AUSGABE <022>
220 PRINT"NULLSTELLE=";Z <071>
230 END <232>
240 REM FUNKTION <154>
250 F=X^2-4 <151>
260 RETURN <064>
270 REM STARTWERTE <051>
280 DATA 3,4 <009>

```

64'er

```

REGULA FALSI
X      F
3      5.00000001
4      12
2.28571428      1.2244898
2.09090909      .371900825
2.00593472      .0237740986
2.00013169      5.26783988E-04
2.00000019      7.76723028E-07
NULLSTELLE= 2

```

Listing 3. Nullstellensuche mit der Regula Falsi

```

100 REM INTEGRATION NACH KEPLER <223>
110 PRINT"KEPLERSCHE FASSREGEL" <212>
120 INPUT"INTERERVALLGRENZEN";A,B <168>
130 IF B<=A THEN PRINT"FALSCH EINGABE!";GOTO 120 <202>
140 REM BERECHNUNG DES GENAHEERTEN INTEGRALWERTES <034>
150 X=A:GOSUB 240 <247>

```

```

160 I=I+F/6 <224>
170 X=B:GOSUB 240 <027>
180 I=I+F/6 <244>
190 X=(A+B)/2:GOSUB 240 <137>
200 I=I+F*2/3:I=I*(B-A) <004>
210 REM AUSGABE <022>
220 PRINT"ERGEBNIS=";I <138>
230 END <232>
240 F=X*X*2 <086>
250 RETURN <052>

```

64'er

```

KEPLERSCHE FASSREGEL
INTERERVALLGRENZEN 0 5
ERGEBNIS= 83.3333333

```

Listing 4. Numerische Integration mit der Kepler'schen Faßregel

```

100 REM INTEGRATION NACH SIMPSON <190>
110 PRINT"SIMPSONSCHE REGEL" <165>
120 INPUT"INTERERVALLGRENZEN";A,B <168>
130 IF B<=A THEN PRINT"FALSCH EINGABE!";GOTO 120 <202>
140 INPUT"TEILINTERVALLE";N <248>
150 H=(B-A)/(2*N):REM SCHRITTWEITE <092>
160 REM BERECHNUNG DES GENAHEERTEN INTEGRALWERTES <054>
170 X=A:GOSUB 330 <138>
180 I=I+F <232>
190 X=B:GOSUB 330 <174>
200 I=I+F <252>
210 FOR J=0 TO N-1 <184>
220 X=A+(2*J+1)*H:GOSUB 330 <103>
230 I=I+4*F <050>
240 NEXT J <076>
250 FOR J=1 TO N-1 <000>
260 X=A+2*J*H:GOSUB 330 <254>
270 I=I+2*F <219>
280 NEXT J <118>
290 I=I*H/3 <034>
300 REM AUSGABE <114>
310 PRINT"ERGEBNIS=";I <230>
320 END <068>
330 F=X*X*2 <178>
340 RETURN <144>

```

64'er

```

SIMPSONSCHE REGEL
INTERERVALLGRENZEN 0 5
TEILINTERVALLE 10
ERGEBNIS= 83.3333334

```

Listing 5. Integration mit der Simpson'schen Regel.

```

100 REM EULERSCHES POLYGONVERFAHREN <002>
110 : <086>
130 PRINT"(CLR,RVSON,5SPACE)EULERSCHES POLYGONZUGVERFAHREN(5SPACE,RVOFF)" <247>
140 PRINT"DIESES VERFAHREN HAT KEINE FEHLER-" <020>
150 PRINT"ABSCHAETZUNG. NUMERISCH IST DIES ES " <148>
160 PRINT"VERFAHREN RECHT SCHNELL, ABER BEI ZU" <072>
170 PRINT"GROSSER SCHRITTWEITE SEHR UNGENAU, WOBEI" <223>
180 PRINT"(UP)DER FEHLER MIT WACHSENDEM X IMMER " <066>
190 PRINT"GROSSER WIRD. SCHRITTWEITEN VON 10^-2" <015>
200 PRINT"UND KLEINER SIND SCHON EINIGERMASSEN" <236>
210 PRINT"GENAU. (DOWN)"" <192>
220 : <196>
230 INPUT"(DOWN)ANFANGSWERTE : X,Y ";X1,Y1 <048>
240 INPUT"(DOWN)INTERVALLOBERGRENZE";B:A=X1 <018>
250 INPUT"(DOWN)SCHRITTWEITE(7SPACE)";H <183>
260 : <238>
280 PRINT"(CLR)Y'=Y*X-2*X (DOWN)" <251>

```

```

290 DEF FN F(X)=INT(X*10E6+.5)/10E6      <035>
300 :                                       <022>
310 Y=Y1                                   <237>
320 PRINT " X(3SPACE);" , " (2SPACE)Y"   <059>
330 FOR X=A TO B STEP H                   <234>
340 GOSUB 410:M=F:REM STEIGUNG DER NAEHERU
    NGSGERADEN                             <146>
350 T=Y-M*(X+A):REM VERSCHUB DER GERADEN <026>
360 PRINT FN F(X),FN F(Y)                 <155>
370 Y=M*(A+X+H)+T:REM NAECHSTER Y-WERT   <007>
380 NEXT X                                 <074>
390 END                                     <138>
400 :                                       <122>
410 F=Y*X-2*X:REM DIFFERENTIALGLEICHUNG  <125>
420 RETURN                                  <224>

```

0 64'er

DIESES VERFAHREN HAT KEINE FEHLER-ABSCHAEZUNG. NUMERISCH IST DIESES VERFAHREN RECHT SCHNELL, ABER BEI ZU GROSSER SCHRITTWEITE SEHR UNGENAU, WOBEI DER FEHLER MIT WACHSENDEM X IMMER GROESSER WIRD. SCHRITTWEITEN VON 10⁻² UND KLEINER SIND SCHON EINIGERMASSEN GENAU.

anfangswerte : x,y 0 1
 intervallobergrenze 2
 schrittweite .2
 y'=y*x-2*x

x	y
0	1
.2	1
.4	.96
.6	.8768
.8	.742016
1	.5407386
1.2	.2488863
1.4	-.171381
1.6	-.7793677
1.8	-1.6687654

Listing 6. Näherungsweise Lösung einer gewöhnlichen Differentialgleichung nach dem Eulerschen Polygonzugverfahren

```

100 PRINT "{CLR,RVSON,SPACE}RUNGE-KUTTA-VER
    FAHREN OHNE LOK. FEHLER{SPACE,RVOFF}" <061>
110 PRINT "DGL AM PROGRAMME EINGEBEN!" <099>
120 INPUT "{DOWN}SCHRITTWEITE {2SPACE}";H <053>
130 INPUT "{DOWN}ANFANGSWERT {3SPACE}";Y0 <133>
140 INPUT "{DOWN}INTERV. GRENZEN";A,B <200>
150 DEF FN R(X)=INT(X*1E4+.5)/1E4 <089>
160 PRINT "{2DOWN,SPACE}X"," Y" <087>
170 YK=Y0 <203>
180 XK=A <194>
190 GOSUB 250 <222>
200 YL=YK+(H/6)*(L1+2*L2+2*L3+L4) <251>
210 PRINT FN R(XK),YK <138>
220 XK=XK+H:YK=YL <216>
230 IF XK<=B+H THEN 190 <199>
240 END <242>
250 X=XK:Y=YK:GOSUB 300:L1=F <107>
260 X=XK+H/2:Y=YK+(H/2)*L1:GOSUB 300:L2=F <031>
270 X=XK+H/2:Y=YK+(H/2)*L2:GOSUB 300:L3=F <042>
280 X=XK+H:Y=YK+H*L3:GOSUB 300:L4=F <213>
290 RETURN <094>
300 F=Y*X-2*X :REM <=> Y'=-... +DGL <038>
310 RETURN <114>

```

0 64'er

DGL AM PROGRAMME EINGEBEN!

schrittweite .2
 anfangswert 1
 interv.grenzen 0 2

x	y
0	1
.2	.979798667
.4	.916713007
.6	.802782992
.8	.622873585
1	.351283323
1.2	-.0544188081
1.4	-.664413992
1.6	-1.59652193
1.8	-3.0527736
2	-5.38822485

Listing 7. Näherungsweise Lösung einer gewöhnlichen Differentialgleichung nach dem Runge-Kutta Verfahren

Vektorrechnung aus dem Effe

Mit Mathefix können fast alle Aufgaben aus dem Bereich der Vektorrechnung im dreidimensionalen Raum gelöst werden.

Alle erdenklichen Schnittprobleme von einer Geraden mit einer Ebene, zwei Geraden oder zwei Ebenen sind ganz einfach zu lösen und benötigen nicht mehr seitenlange Berechnungen. Auch Schnittwinkel, Projektion von Punkten und Umrechnungen von der Parameterform in die Normalenform der Ebene stellen kein Problem mehr dar. Jede Ausgabe erfolgt, sinnvoll auf zwei Stellen gerundet, auf dem Bildschirm, und kann mit der F1-Taste als Hardcopy auf jedem Drucker ausgegeben werden, der sich von Simons Basic ansprechen läßt. Das Programm arbeitet durchgehend mit HiRes-Grafik.

Wichtige Eingabehinweise:

Das Programm (Listing) benötigt Simons Basic. Um den C 64 ebenfalls auf Kleinschrift umzustellen, geben Sie im Direktmodus den Befehl »CSET 1«. Beachten Sie bitte den REM-Kommentar in Zeile 15. Ab Zeile 1020 steht am Anfang der Strings öfter ein kleines reverse »b«. Es entsteht durch die Tastenkombination »CTRL« + »B«. Das reverse »H« (zum Beispiel in Zeile 1441) ist die F7-Taste.

Allgemeine Hinweise:

In allen Menüs genügt das Drücken der gewünschten Taste um in das nachfolgende Menü zu kommen oder die Berechnung zu beginnen. Auf die Aufforderung »WEITER MIT RETURN« und »ZURÜCK MIT RETURN« drücken Sie bitte die RETURN-Taste. Bei der Eingabe Ihrer Zahlenwerte schließen Sie bitte jede Zahleneingabe mit einem RETURN ab.

Durch das Betätigen der Funktionstaste F7 innerhalb der Werteingabe gelangen Sie jeweils in das vorherige Menü zurück.

Bei »ZURÜCK MIT RETURN« können Sie durch drücken der SPACE-Taste dieselbe Rechnung mit anderen Zahlenwerten wiederholen. Jedesmal, wenn von Ihnen eine Eingabe verlangt wird, können Sie mit der Funktionstaste F1 eine Hardcopy des aktuellen Bildschirmes auf dem Drucker erzeugen.

Im Programm wird unter »n« der Punkt mit den Koordinaten (X/Y/Z) verstanden.

Das Programm enthält eine Anleitung in Kurzform.